In some experiments, you randomize treatments on multiple "levels" of experimental units, commonly referred to as the *whole units* and *sub units* (in field trials, *whole plots* and *sub plots*). The sub units are nested within the whole units, thus whole units are large and sub units are small. Each level is expected to have a different variance, larger units (whole units) generally having greater variance than smaller units (sub units). Two sizes of units require that two random experimental variances be estimated. This type of experiment is a mixed model, since whole units and sub units are considered random. The treatment structure for a split plot is a factorial since one (or more) factor is assigned to the whole units and one (or more) factor is assigned to the sub units. The sub unit design is always a block design, while the whole unit design may be any of the designs that we have studied (e.g., CR, RCB, LS, etc.). Therefore the complete design name for split plot also includes the identification of the whole plot design. For example, if the whole plot treatment factor is assigned as a RCB then the design would be called a Randomized Complete Block Split Plot Design.

MIXED syntax for Split Plot Analysis:

PROC MIXED ratio covtest;
CLASS <classification variables>**;**
MODEL <dep var> = <fixed sources>**;**

Specify the fixed sources of variation. In most cases these will be the sources of variation representing the factorial treatment structure.

RANDOM <random sources>**;**

Specify the random sources of variation. In most cases this will be any blocking factors, plus the whole and sub unit error terms. It is the correct identification of the whole and sub unit error terms that make the analysis a split plot.

LSMEANS <fixed sources>**/pdiff;**
ESTIMATE . . .
CONTRAST . . .

The MIXED procedure uses the expected mean squares to compute the correct standard error of the difference for any of the mean comparison procedures, ESTIMATE, and CONTRAST statements.

ODS . . .

SAS program and listing split plot example

```
1  TITLE1 'LAB#9:  ANALYZING SPLIT PLOT DESIGNS';
2
3  OPTIONS LS=66 PS=54 PAGENO=1;
4
5  TITLE2 'The Whole Plot Design is a Randomized Complete Block';
6  TITLE3 'The Whole Plot Treatment is Irrigated vs Non-irrigated';
7  TITLE4 'The Sub Plot Trt is Level of Nitrogen (0, 40, 80, 160)';
8  TITLE5 'The data are expressed in yield per hectare';
9  DATA sp;
10 INPUT blk ig$ nit yield;
11 LINES;

       Data lines entered here

45 RUN;
```

NOTE: The data set WORK.SP has 32 observations and 4 variables.

COMMENTS

This experiment is a randomized complete block split plot with four blocks. The whole plot treatment is irrigated or non-irrigated and the sub plot treatment is level of nitrogen (0, 40, 80 or 160), making the treatment structure a 2x4 factorial. The dependent variable is crop yield per hectare.

```
47   TITLE5 'Print of data file';
48   PROC PRINT DATA=sp;
49   QUIT;
```

```
            LAB#9:  ANALYZING SPLIT PLOT DESIGNS                1
     The Whole Plot Design is a Randomized Complete Block
    The Whole Plot Treatment is Irrigated vs Non-irrigated
  The Sub Plot Treatment is Level of Nitrogen (0, 40, 80, 160)
                      Print of data file

                OBS    BLK    IG    NIT    YIELD

                  1      1    N       0       26
                  2      1    N      40       31
                  3      1    N      80       33
                  4      1    N     160       25
                  5      1    Y       0       32
                  6      1    Y      40       41
                  7      1    Y      80       49
                  8      1    Y     160       46
                  9      2    N       0       22
                 10      2    N      40       29
                 11      2    N      80       35
                 12      2    N     160       24
                 13      2    Y       0       31
                 14      2    Y      40       29
                 15      2    Y      80       38
                 16      2    Y     160       44
                  .
                  .
                  .
                 25      4    N       0       13
                 26      4    N      40       24
                 27      4    N      80       20
                 28      4    N     160       15
                 29      4    Y       0       20
                 30      4    Y      40       25
                 31      4    Y      80       33
                 32      4    Y     160       32
```

```
51   TITLE5 'Mixed model analysis of variance';
52   TITLE6 'Orthogonal polynomial contrasts, Nit and IG*Nit
            interaction';
53   PROC MIXED DATA=sp RATIO COVTEST;
54   CLASS blk ig nit;
55   MODEL yield = ig nit ig*nit;
56   RANDOM blk blk*ig;
57   CONTRAST 'Nitrogen linear'     nit -7 -3  1  9;
58   CONTRAST 'Nitrogen quadratic'  nit  7 -4 -8  5;
59   CONTRAST 'Nitrogen cubic'      nit -3  8 -6  1;
60   CONTRAST 'Ig*Nit linear'    ig*nit -7 -3  1  9  7  3 -1 -9;
61   CONTRAST 'Ig*Nit quadratic' ig*nit  7 -4 -8  5 -7  4  8 -5;
62   CONTRAST 'Ig*Nit cubic'     ig*nit -3  8 -6  1  3 -8  6 -1;
63   LSMEANS ig nit ig*nit / PDIFF;
64   ODS exclude listing lsmeans;
65   ODS output LSMEANS=lsm;
66   ODS exclude listing diffs;
67   ODS output DIFFS=diffs;
66   QUIT;
```

```
NOTE: The data set WORK.LSM has 14 observations and 8 variables.
NOTE: The data set WORK.DIFFS has 35 observations and 10
      variables.
```

The MODEL statement identifies the fixed sources of variation. In this case the 2x4 factorial structure. The RANDOM statement identifies the random sources of variation, which are block and the block*irrigation interaction. The block*irrigation interaction is the whole plot MSe, while the residual is the sub plot MSe. Since the nitrogen treatment is quantitative, I have included a number of orthogonal polynomial contrasts. These contrasts should identify the form of the regression equation that would best describe these data. Because of the multiple random variances, it is difficult to correctly analyze data from split plot designs using simple regression techniques.

LAB#9: ANALYZING SPLIT PLOT DESIGNS 2

The Whole Plot Design is a Randomized Complete Block
The Whole Plot Treatment is Irrigated vs Non-irrigated
The Sub Plot Treatment is Level of Nitrogen (0, 40, 80, 160)
Mixed model analysis of variance
Orthogonal polynomial contrasts, Nit and IG*Nit interaction

The Mixed Procedure

Model Information

Data Set	WORK.SP
Dependent Variable	yield

Class Level Information

Class	Levels	Values
blk	4	1 2 3 4
ig	2	N Y
nit	4	0 40 80 160

Dimensions

Observations Used	32

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	175.73630999	
1	1	151.89110756	0.00000000

Convergence criteria met.

LAB#9: ANALYZING SPLIT PLOT DESIGNS 3

The Whole Plot Design is a Randomized Complete Block
The Whole Plot Treatment is Irrigated vs Non-irrigated
The Sub Plot Treatment is Level of Nitrogen (0, 40, 80, 160)
Mixed model analysis of varaince
Orthogonal polynomial contrasts, Nit and IG*Nit interaction

Covariance Parameter Estimates

Cov Parm	Ratio	Estimate	Standard Error	Z Value	Pr Z
blk	3.1785	36.7292	34.5144	1.06	0.1436
blk*ig	0.6532	7.5486	8.5764	0.88	0.1894
Residual	1.0000	11.5556	3.8519	3.00	0.0013

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
ig	1	3	13.04	0.0365
nit	3	18	17.86	<.0001
ig*nit	3	18	4.98	0.0109

Contrasts

Label	Num DF	Den DF	F Value	Pr > F
Nitrogen linear	1	18	24.93	<.0001
Nitrogen quadratic	1	18	28.59	<.0001
Nitrogen cubic	1	18	0.05	0.8298
Ig*Nit linear	1	18	13.12	0.0020
Ig*Nit quadratic	1	18	0.79	0.3860
Ig*Nit cubic	1	18	1.02	0.3252

These are the variance component (VC) estimates for the random effects. In this case the block VC is the largest, followed by the sub plot VC and then the whole plot VC. Remember these are variance components and not MS for the whole unit or block. In fact the whole unit variance (MS) would be the sub unit VC + 4 * whole unit VC [MSw = 11.56 + 4(7.55)=41.76], where the four is the number of sub units per whole unit.

These are the tests of hypotheses for the fixed effects. Note that although the F ratio is quite large for irrigation, it is barely significant due to the small number of degrees of freedom. This is the price the researcher frequently pays for using a split plot design; a large reduction in the sensitivity for tests concerning the whole plot treatment factor. On the other hand, tests at the sub plot level are sometimes more sensitive.

The polynomial contrasts indicate that the nitrogen response contains both a linear and quadratic component and that the two response lines differ in the linear component (slope).

```
68   TITLE5 'Print of main effect and treatment means';
69   PROC PRINT DATA=lsm;
70   FORMAT Estimate StdErr 6.1;
71   VAR EFFECT IG NIT Estimate StdErr;
72   QUIT;
```

LAB#9: ANALYZING SPLIT PLOT DESIGNS 4

The Whole Plot Design is a Randomized Complete Block

The Whole Plot Treatment is Irrigated vs Non-irrigated

The Sub Plot Treatment is Level of Nitrogen (0, 40, 80, 160)

Print of main effects and treatment means

Obs	Effect	ig	nit	Estimate	StdErr
1	ig	N	_	23.9	3.43
2	ig	Y	_	32.1	3.43
3	nit		0	20.7	3.40
4	nit		40	28.7	3.40
5	nit		80	32.5	3.40
6	nit		160	30.0	3.40
7	ig*nit	N	0	18.2	3.74
8	ig*nit	N	40	27.0	3.74
9	ig*nit	N	80	28.0	3.74
10	ig*nit	N	160	22.2	3.74
11	ig*nit	Y	0	23.2	3.74
12	ig*nit	Y	40	30.5	3.74
13	ig*nit	Y	80	37.0	3.74
14	ig*nit	Y	160	37.7	3.74

These are the broad sense estimates of the standard errors of the treatment means. That is they are estimates of the variation that would be expected in repetitions of the experiment based on random samples of other blocks from the same population of blocks as was sampled for the current experiment.

```
74   TITLE5'Print of differences and tests of differences';
75   PROC PRINT DATA=diffs;
76   FORMAT DIFF StdErr 6.1 Probt 6.3;
77   WHERE ig=_IG or nit=_NIT;
78   VAR Effect ig nit _IG _NIT DIFF StdErr DF Probt;
79   QUIT;
```

LAB#9: ANALYZING SPLIT PLOT DESIGNS 5

The Whole Plot Design is a Randomized Complete Block
The Whole Plot Treatment is Irrigated vs Non-irrigated
The Sub Plot Treatment is Level of Nitrogen (0, 40, 80, 160)
Print of differences and tests of differences

Obs	Effect	ig	nit	_ig	_nit	Estimate	StdErr	DF	Probt
1	ig	N	_	Y	_	-8.2	2.3	3	0.036
2	nit		0		40	-8.0	1.7	18	0.000
3	nit		0		80	-11.8	1.7	18	0.000
4	nit		0		160	-9.3	1.7	18	0.000
5	nit		40		80	-3.8	1.7	18	0.041
6	nit		40		160	-1.3	1.7	18	0.472
7	nit		80		160	2.5	1.7	18	0.159
8	ig*nit	N	0	N	40	-8.8	2.4	18	0.002
9	ig*nit	N	0	N	80	-9.8	2.4	18	0.001
10	ig*nit	N	0	N	160	-4.0	2.4	18	0.113
11	ig*nit	N	0	Y	0	-5.0	3.1	18	0.123
15	ig*nit	N	40	N	80	-1.0	2.4	18	0.682
16	ig*nit	N	40	N	160	4.7	2.4	18	0.064
18	ig*nit	N	40	Y	40	-3.5	3.1	18	0.272
21	ig*nit	N	80	N	160	5.8	2.4	18	0.028
24	ig*nit	N	80	Y	80	-9.0	3.1	18	0.009
29	ig*nit	N	160	Y	160	-15.5	3.1	18	0.000
30	ig*nit	Y	0	Y	40	-7.3	2.4	18	0.007
31	ig*nit	Y	0	Y	80	-13.8	2.4	18	0.000
32	ig*nit	Y	0	Y	160	-14.5	2.4	18	0.000
33	ig*nit	Y	40	Y	80	-6.5	2.4	18	0.015
34	ig*nit	Y	40	Y	160	-7.3	2.4	18	0.007
35	ig*nit	Y	80	Y	160	-0.7	2.4	18	0.759

Note that there are four different standard errors of the differences. These reflect the comparisons of means that are based on whole plot variance, sub plot variance or a combination of the two, as well as differences in replication between main effect means and two-way means. If one uses the polynomial contrasts to interpret the nitrogen effect then these mean comparisons should only be used to examine the differences between irrigated and non-irrigated at a given nitrogen level.

```
81   DATA lsm;
82   SET lsm;
83   IF Effect='ig*nit';
84   nitrogen = nit*1;
85   RUN;
```

The ODS statement in the MIXED procedure generated an output data set containing the treatment means. Only the interaction means (IG*NIT) are retained, as these means will be used to plot the data.

```
NOTE: There were 14 observations read from the dataset WORK.LSM.
NOTE: The data set WORK.LSM has 8 observations and 9 variables.
```

```
87   TITLE5 'Plot of trt means for examination of interaction effect';
88   PROC PLOT DATA=lsm VPERCENT=70;
89   FORMAT Estimate 5.0;
90   PLOT Estimate*nitrogen=ig / VAXIS=0 TO 40 BY 10;;
91   QUIT;
```

The PLOT procedure is used to construct a crude plot of the treatment means to examine the irrigation by nitrogen interaction. The VAXIS option determines the placement of tick marks on the vertical axis.



The interaction can be seen as an increasing difference between irrigated and non-irrigated as the nitrogen level increases. In addition, very little increase in yield is seen as nitrogen is increased from 80 to 160 units under irrigation, while yields of non-irrigated plots decreased at the highest level of nitrogen.